


3.1 ESEM MIMIC

Example: ESEM MIMIC



- Step 1: Unrotated loadings model, estimating Λ with $\Psi = I$ (standardized, uncorrelated factors), and the Γ regressions of factors on covariates
- Step 2: EFA rotation of the y measurement part gives new Λ , Ψ , and a corresponding transformation is applied to the estimated Γ from Step 1

Slide 68 shows an example with an EFA measurement model that has 2 factors and 6 indicators, to which we add 2 covariates x_1 and x_2 . The loading matrix is called Λ , the factor covariance matrix is called Ψ and the regression matrix is called Γ . Ψ is now actually a residual covariance matrix for the factors. This is called a MIMIC model where MIMIC stands for multiple indicators, multiple causes.

Typically, MIMIC modeling is done by using a CFA measurement model but here the ESEM approach makes it possible to use an EFA measurement model. ESEM uses two steps:

In Step 1, an unrotated loadings model estimates Λ with $\Psi = I$ (so standardized, uncorrelated factors), and the Γ regressions of factors on covariates.

In Step 2, an EFA rotation is made of the y measurement part and gives new Λ and Ψ , after which a transformation is applied to the estimated Γ from Step 1.

ESEM MIMIC Inputs

ESEM MIMIC

MODEL:

```
f1-f5 BY y1-y15 (*1);  
f1-f5 ON age;
```

ESEM MIMIC with rescovs

MODEL

```
f1-f5 BY y1-y15 (*1);  
y1 y5 y9 y12 WITH y1 y5 y9 y12;  
f1-f5 ON age;
```

Bi-Factor ESEM MIMIC

ANALYSIS:

```
ROTATION = BI-GEOMIN;
```

MODEL:

```
fg f1-f4 BY y1-y15(*1);  
fg-f4 ON age;
```

Slide 69 shows the generality of the ESEM MIMIC approach in three variations of the input: the EFA MIMIC just discussed, a version that adds some residual covariances among the indicators, and MIMIC with a bifactor measurement model where the general factor is also regressed on the covariates.

ESEM MIMIC Examples

- User's Guide ex 5.24: ESEM
 - f1-f2 BY y1-y8 (*1); f1-f2 ON x1-x2;
- Asparouhov-Muthén (2026): PSEM SEFA for HS19 with second-order factor regressed on grade and gender
- Muthén (2025): ESEM and PSEM with WLSMV for binary factor indicators of antisocial behavior
 - ESEM MIMIC analysis: EFA measurement model with factors regressed on covariates and covariates influencing indicators directly

Slide 70 lists some applications of ESEM MIMIC.

Apart from the User's Guide example 5.24, our 2026 paper illustrates the use of a SEFA measurement model in a PSEM analysis where the second-order factor is regressed on covariates.

The Muthén 2025 handbook chapter shows ESEM together with PSEM with WLSMV estimation for binary indicators. Both the factors and the indicators are regressed on covariates. The indicator regressions capture measurement non-invariance. This example will be shown next.

Input for ESEM MIMIC with 17 Items, 3 Factors, and 11 Covariates: ASB Example (Muthén, 2025)

```
VARIABLE:      CATEGORICAL = property-gambling;

ANALYSIS:      ESTIMATOR = WLSMV;

MODEL:         ! 3-factor EFA measurement model:
               f1-f3 BY property-gambling (*1);

               ! EFA factors regressed on covariates:
               f1-f3 ON sex black hisp single divorce dropout
               college onset age94 dep abuse;

OUTPUT:        TECH1 TECH4 TECH10
               STANDARDIZED RESIDUAL;

PLOT:          TYPE = PLOT3;
```

Slide 71 shows the input used in the 2025 paper. The example refers to antisocial behavior where 17 binary variables are used as indicators of 3 factors regressed on 11 covariates.

The VARIABLE command specifies the indicators as categorical and the ANALYSIS command requests weighted least squares estimation. The EFA specification uses the ESEM approach, followed by the regression of the EFA factors on the covariates and using the 2 rotation steps outlined earlier on slide 68.

- ESEM-PSEM analysis: Direct effects added from covariates to factor indicators to capture measurement non-invariance
 - Using PSEM ALF priors makes it possible to identify all direct effects
 - The ALF prior variance is chosen so that the loglikelihood is the same as the "null" model which in this case is ESEM with regressions of all indicators on all covariates but no regressions of the factors on the covariates
- 17 factor indicators and 11 covariates results in 187 possible direct effects, of which PSEM finds that 40 are significant
 - Freeing these 40 direct effects changes the relations between the factors and the covariates

Slide 72 discusses the addition of direct effects from the covariates to the indicators. The presence of such effects reflect measurement non invariance.

The direct effects are not identified when added all at the same time, but PSEM using ALF makes it possible to identify them.

Here, it is important to choose the prior variance of ALF properly. It should be chosen such that the loglikelihood is the same as the so called null model which in this case is the model with all direct effects but no effects on the factors.

There is a total of 187 possible direct effects. The PSEM ALF approach avoids a long stepwise search for significant effects and includes all of them in one single run. In this case, 40 effects were significant. These effects can then be freed in a new run to check to which extent the relations have changed between the factors and the covariates. The paper, which is on the Mplus website, gives further details and discusses the interpretation of the findings.

The input is shown on the next slide.

Input for PSEM Direct Effects for ASB

```
USEVARIABLES = property-gambling sex black hisp single
divorce dropout college onset age94 dep abuse;
CATEGORICAL = property-gambling;

ANALYSIS: ESTIMATOR = WLSMV;

MODEL: f1-f3 BY property-gambling (*1);
f1-f3 ON sex black hisp
single divorce dropout
college onset age94 dep abuse;
property-gambling ON sex-abuse (p1-p187);

MODEL PRIORS: p1-p187 ALF(0,1);

OUTPUT: TECH1 TECH4 TECH10
STANDARDIZED RESIDUAL;

PLOT: TYPE = PLOT3;
```

Slide 73 gives the input for the analysis just discussed. The bolded lines show the changes made to the regular ESEM MIMIC. The direct effects are labeled and then given ALF priors with mean zero and variance 1.